Example 2.17 CSTR

Problem Statement

Consider the following hypothetical complex reaction scheme for a liquid phase system:

$$A \xrightarrow{r_1} 2B$$

$$A \underset{r_3}{\overset{r_2}{\rightleftarrows}} C$$

$$B \xrightarrow{r_4} D + C$$

where

$$r_1 = k_1 C_A$$
$$r_2 = k_2 C_A^{1/2}$$
$$r_3 = k_3 C_C^2$$
$$r_4 = k_4 C_B^2$$
$$k_1 = 1.0 \text{ sec}^{-1}$$
$$k_2 = 0.2 \text{ liter}^{1/2} / \text{gmole}^{1/2}\text{-sec}$$
$$k_3 = 0.05 \text{ liter/gmole-sec}$$
$$k_4 = 0.4 \text{ liter/gmole-sec}$$

where

r_i [=] gmoles/liter-sec

A continuously stirred tank reactor (CSTR) is used for this reaction system (**Figure 2.19**). The reactor volume, V_R, is 100 liters and the volumetric feed to the reactor Q is 50 liters/sec at a concentration of 1.0 gmole/liter of component A.

Since a CSTR is designed to operate at steady state and this system is assumed to be operated under isothermal conditions, steady-state mole balances define the performance of this system (i.e., concentrations coming out of the reactor). The following mole balances can be constructed:

	OUT = IN	+ GENERATION	− CONSUMPTION
(Component A)	$C_A Q = C_{A_0} Q$	$+ V_R(r_3)$	$- V_R(r_1 + r_2)$
(Component B)	$C_B Q = 0$	$+ V_R(2r_1)$	$- V_R(r_4)$
(Component C)	$C_C Q = 0$	$+ V_R(r_2 + r_4)$	$- V_R(r_3)$
(Component D)	$C_D Q = 0$	$+ V_R(r_4)$	$- 0$



Figure 2.19 CSTR for Example 2.17.

Substituting the rate expressions and rearranging results in the following set of nonlinear equations:

$$f_1(x) = -C_A + C_{A_0} + V_R(-k_1 C_A - k_2 C_A^{3/2} + k_3 C_C^2)/Q = 0$$
$$f_2(x) = -C_B + V_R(2k_1 C_A - k_4 C_B^2)/Q = 0$$
$$f_3(x) = -C_C + V_R(k_2 C_A^{3/2} - k_3 C_C^2 + k_4 C_B^2)/Q = 0$$
$$f_4(x) = -C_D + V_R(k_4 C_B^2)/Q = 0$$

Note that the only unknowns are C_A, C_B, C_C, and C_D. Solve these equations using Newton's method.

Solution

In order to implement Newton's method, the partial derivatives of each equation with respect to each variable (the Jacobian) must be generated. Listed are the nonzero elements of the Jacobian for this problem:

$$\frac{\partial f_1(x)}{\partial C_A} = -1 - \frac{V_R\,(k_1 + 1.5k_2\,C_A^{1/2})}{Q}$$
$$\frac{\partial f_1(x)}{\partial C_C} = \frac{V_R\,(2k_3\,C_C)}{Q}$$
$$\frac{\partial f_2(x)}{\partial C_A} = \frac{V_R\,(2k_1)}{Q}$$
$$\frac{\partial f_2(x)}{\partial C_B} = -1 - \frac{V_R\,(2k_4\,C_B)}{Q}$$
$$\frac{\partial f_3(x)}{\partial C_A} = \frac{V_R\,(1.5k_2\,C_A^{1/2})}{Q}$$

$$\frac{\partial f_3(x)}{\partial C_B} = \frac{V_R\,(2k_4\,C_B)}{Q}$$

$$\frac{\partial f_3(x)}{\partial C_C} = -1 - \frac{V_R\,(2k_3\,C_C)}{Q}$$

$$\frac{\partial f_4(x)}{\partial C_B} = \frac{V_R\,(2k_4\,C_B)}{Q}$$

$$\frac{\partial f_4(x)}{\partial C_D} = -1$$

These equations and the system of nonlinear equations presented in the problem statement can be used in conjunction with **Equation 2.33** and **2.34** to effect a numerical solution of the problem. Assume the following values for the starting point: $C_A = 0.3$, $C_B = 0.3$, $C_C = 0.3$, and $C_D = 0.3$

Following is a program for the solution of this problem along with the results.

PROGRAM LISTING FOR EXAMPLE 2.17

```
C*************************   ABSTRACT   ********************************
C
C     THIS PROGRAM CALCULATES THE PERFORMANCE OF AN ISOTHERMAL CSTR
C  REACTOR. USING A STEADY STATE MOLE BALANCE FOR EACH REACTION
C  SPECIES, A SYSTEM OF FOUR NONLINEAR EQUATIONS CONTAINING FOUR
C  UNKNOWNS IS GENERATED.  THIS SYSTEM OF EQUATIONS IS SOLVED USING
C ,THE NEWTON METHOD.
C
C*************************   NOMENCLATURE   ********************************
C
C  CAO- THE INITAL CONCENTRATION OF A IN THE FEED STREAM (GMOLES/L)
C  ERRIM- ERROR CRITERIA
C  FX(I)- THE VALUE OF THE NONLINEAR EQUATIONS WHERE I INDICATES
C         WHICH EQUATION
C  KI- THE RATE CONSTANT FOR EACH OF THE REACTIONS (I INDICATES THE
C      THE REACTION NUMBER)
C  N- THE NUMBER OF NONLINEAR EQUATIONS
C  Q- THE VOLUMETRIC FLOW RATE INTO AND OUT OF THE REACTOR (L/SEC)
C  X(I)- THE INDEPENDENT VARIABLES OF THE PROBLEM  I=1 (CONC OF A),
C        I=2 (CONC OF B), I=3 (CONC OF C), AND I=4 (CONC OF D)
C        ALL CONC ARE GIVEN IN GMOLES/L
C
C
C*****************   INPUT DESCRIPTION   ********************
C
C     THE INITIAL GUESSES ARE SPECIFIED  IN THE MAIN PROGRAM AS WELL
C  AS THE ERROR CRITERIA AND THE NUMBER OF NONLINEAR EQUATIONS.
C  THE FUNCTIONS ARE SPECIFIED IN SUBROUTINE FUNC AND THE PARTIAL
C  DERIVATIVES OF THE FUNCTION WITH RESPECT TO THE INDEPENDENT
C  VARIABLES ARE SPECIFIED IN SUBROUTINE DER.
C
C
      IMPLICIT REAL*8(A-H,O-Z)
```

```
      DIMENSION X(4),FX(4),Y(4),A(4,4),B(4),D(4),IPVT(4)
      COMMON /ONE/ K1,K2,K3,K4,VR,Q,CAO
C  SPECIFY EXTERNAL STATEMENT
      EXTERNAL FUNC,DER
      REAL K1,K2,K3,K4
C
C  INPUT THE RATE CONSTANTS
C
      K1=1.
      K2=.2
      K3=.05
      K4=.4
C
C  INPUT REACTOR VOLUME, FLOW RATE, AND INLET CONCENTRATION
C
      VR=100.
      Q=50.
      CAO=1.
C
C  MAKE INITIAL GUESSES
C
      X(1)=.3
      X(2)=.3
      X(3)=.3
      X(4)=.3
      N=4
      ERLIM=1.D-2
C
C  CALL NEWTON METHOD
C
      CALL NNEWTN(FUNC,DER,N,X,FX,ERLIM,A,B,D,IPVT)
C
C  PRINT OUT FINAL RESULTS
C
      DO 6 I=1,4
    6 WRITE(6,7)I,X(I)
    7 FORMAT(' 10X,3H I=,I3,5X,3H X=,D14.7)
      STOP
      END


C
C************************  ABSTRACT  ********************************
C
C     THIS SUBROUTINE CALCULATES THE PARTIAL DERIVATIVES OF THE
C  THE FUNCTIONS WITH RESPECT TO THE INDEPENDENT VARIABLES.
C  A(I,J) REPRESENTS THE PARTIAL OF THE ITH FUNCTION WITH RESPECT
C  TO THE JTH VARIABLE.
C
C************************  INPUT DESCRIPTION  ***********************
C
C     THE VALUE OF N AND X(I) ARE SUPPLIED TO SUBROUTINE DER BY
C  SUBROUTINE
C  NNEWTN THROUGH THE CALLING STATEMENT.
C
C*******************************************************************
C
```

```
C
      SUBROUTINE DER(N,X,A)
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION A(4,4),X(4)
      REAL*8 K1,K2,K3,K4
      COMMON /ONE/K1,K2,K3,K4,VR,Q,CAO
      DO 1 I=1,N
      DO 1 J=1,N
    1 A(I,J)=0.0
      A(1,1)=-1.-VR*K1/Q-1.5*K2*VR*X(1)**.5/Q
      A(1,3)=VR*2.*K3*X(3)/Q
      A(2,1)=2.*VR*K1/Q
      A(2,2)=-1-2.*K4*X(2)/Q*VR
      A(3,1)=1.5*VR*K2*X(1)**.5/Q
      A(3,2)=2.*K4*X(2)/Q*VR
      A(3,3)=-1.-2.*K3*X(3)/Q*VR
      A(4,2)=2.*VR*K4*X(2)/Q
      A(4,4)=-1.
      RETURN
      END
```

```
C
C*****************************  ABSTRACT  ******************************
C
C     THIS SUBROUTINE CALCULATES THE VALUES OF EACH NONLINEAR
C  EQUATION GIVEN THE VALUE OF X(I) AND N.  THESE VALUES ARE
C  SUPPLIED TO THIS SUBROUTINE WHEN IT IS CALLED BY SUBROUTINE
C  NNEWTN.
C
C************************************************************************
C
      SUBROUTINE FUNC(N,X,FX)
      IMPLICIT REAL*8(A-H,O-Z)
      COMMON /ONE/K1,K2,K3,K4,VR,Q,CAO
      DIMENSION X(4),FX(4)
      REAL*8 K1,K2,K3,K4
      FX(1)=CAO  -X(1)+VR*(-K1*X(1)+K3*X(3)*X(3)-K2*X(1)**1.5)/Q
      FX(2)=-X(2)+VR*(2.*K1*X(1)-K4*X(2)*X(2))/Q
      FX(3)=-X(3)+VR*(K2*X(1)**1.5-K3*X(3)*X(3)+K4*X(2)*X(2))/Q
      FX(4)=-X(4)+VR*(K4*X(2)*X(2))/Q
      WRITE(6,11)(X(I),FX(I),I=1,N)
   11 FORMAT(' 10X,' X=',D14.7,5X,' F=',D14.7)
      WRITE(6,22)
   22 FORMAT( /)
      RETURN
      END
```

```
C
C*****************************  ABSTRACT  ******************************
C
C     THIS SUBROUTINE EMPLOYES NEWTON'S METHOD IN ORDER TO SOLVE A
C  SET OF N NONLINEAR EQUATIONS CONTAINING N UNKNOWNS.
C  THIS SUBROUTINE IS CALLED BY THE MAIN PROGRAM AND IS SUPPLIED
C  THE VALUES OF THE INITIAL GUESS FOR X(I)'S AS WELL AS THE VALUE OF
C  N.  THIS SUBROUTINE USES THE VALUES OF THE FUNCTION FROM FUNC AND
```

```
C  THE VALUES OF THE PARTIAL DERIVATIVES OF THE FUNCTION IN ORDER TO
C  DETERMINE THE SOLUTION. THIS METHOD USES THE LIBRARY ROUTINE LINPAC
C  TO SOLVE THE SYSTEM OF LINEAR EQUATIONS USED BY NEWTON'S METHOD.
C
C*********************************************************************
C
C
      SUBROUTINE NNEWTN(FUNC,DER,N,X,FX,ERLIM,A,B,D,IPVT)
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION A(N,1),B(1),D(1),IPVT(1),X(1),FX(1)
      EXTERNAL FUNC,DER
    1 CONTINUE
      ITEST=0
C
C  MAKE FUNCTION EVALUATIONS
C
      CALL FUNC(N,X,FX)
      DO 3 I=1,N
    3 B(I)=-FX(I)
C
C  EVALUATE JACOBIAN MATRIX
C
      CALL DER(N,X,A)
C
C  CALL LINEAR EQUATION SOLVER
C
      CALL LINPAC(N,A,B,D,IPVT)
C
C  MAKE AN IMPROVED VALUE FOR X(I)
C
      ITEST=0
      DO 5 I=1,N
C  ANTICIPATE ZERO ROOTS
      TS=DABS(X(I))
      IF(TS.LT.1.D-10)GO TO 5
      RAT=B(I)/X(I)
      IF(DABS(RAT).GT.ERLIM)ITEST=ITEST+1
      IF (DABS (D(I)/X(I)).GT.0.5)D(I)=0.5*X(I)*DABS(D(I))/B(I)
    5 X(I)=X(I)+D(I)
C
C  CHECK FOR CONVERGENCE
C
      IF(ITEST.NE.0)GO TO 1
      RETURN
      END
```

```
D=  .3000000D+00      F=  .4327325D-01
D=  .3000000D+00      F=  .8280000D-06
D=  .3000000D+00      F= -.1712733D-06
D=  .1000000D+00      F= -.2280000D-09
```

```
D=  .3150711D+00      F=  .1258026D-02
D=  .9001920D+00      F= -.2881844D+00
D=  .4148790D+00      F=  .2869264D+00
D=  .3600922D+00      F=  .2881844D+00

D=  .3183329D+00      F=  .1342964D-02
D=  .7874451D+00      F= -.1016950D-01
D=  .5308878D+00      F=  .8826533D-02
D=  .4858863D+00      F=  .1016950D-01

D=  .3188652D+00      F=  .1597539D-05
D=  .7838874D+00      F= -.1012589D-04
D=  .5349778D+00      F=  .8528346D-05
D=  .4915734D+00      F=  .1012589D-04

I=  1       X=  .3198658D+00
I=  2       X=  .7838840D+00
I=  3       X=  .5349819D+00
I=  4       X=  .4915793D+00
```

Program Discussion: Note that only four iterations are required for convergence. Also note that the convergence was quite rapid near the end (i.e., quadratic convergence). This problem could have been solved by considering only three unknowns. Do you see why this is true?

Figure 2.20 shows converged values of concentration of A, B, C, and D for different values of Q, the volumetric feed to the reactor. This problem becomes quite nonlinear for Q less than 12.0 liters/sec. Above 12.0 liters/sec, solutions were obtained relatively easily using the original guesses for concentrations, i.e., $C_A = C_B = C_C = C_D = 0.3$. Even using the converged solutions for $Q = 11.4$ liters/sec as the initial guesses for $Q = 11.3$ liters/sec did not result in a converged solution. Using Newton's method for $Q = 11.3$ liters/sec resulted in negative values for the concentrations (extraneous roots).

For $Q < 12.0$ there is generation of all the molar species, which is not physically realistic. This can be understood by analysis of the reaction stoichiometry. The first reaction produces $2B$, the second reaction produces C, and the third reaction can convert C back to A. This is an example of an *autocatalytic* reaction scheme since the reactant A reacts to produce more A and so on. **Example 2.21**, the application of the tearing method to this problem, will shed more light on the observed difficulties.



Figure 2.20 Additional results for **Example 2.17**.